
Prolog and Relational Databases

Similarities: analogies between RDB and PROLOG

z records are PROLOG facts,

z tables are PROLOG collections of facts,

z links can be representad as binary facts,

z associative tables can be representad as collections binary facts,

z SELECT – easily implemented by backtracking search and pattern matching,

z all relational/set operations can be easily mimic with PROLOG constructs.

Dissimilarities: advantages of PROLOG

z terms and unification – structural objects and matching them accessible,

z lists – no problem with multiple values,

z clauses – inference instead of pure search,

z recursion – intrinsic in PROLOG,

z iteration – simple in PROLOG,

z repeat-fail – built-in backtracking search,

z operators – custom knowledge representation language creation,

z meta-programming – a unique feature;

PROLOG vs. SQL

Problems with SQL

SQL is not a complete language!
SQL has no recursion.

Example problems with SQL

z variables — lack of variables,

z transitive cosure,

z recurency,

z lists,

z structures,

z trees,

z graphs,

z loops,

z combinatorial probelms,

z constraint satisfaction,

z search,

z backtracking search.

Two Example Problems

Shopping: a Knapsack Problem

1 item(item1,123).
2 item(item2,234).
3 item(item3,345).
4 item(itme4,456).
5 item(item5,567).

Task: find all sets of items that can be purchased
for a given total cost.

Travelling: Path Generation

1 link(a,b).
2 link(a,c).
3 link(b,c).
4 link(b,d).
5 link(c,d)
6 link(c,e).
7 link(d,e).

Task: find all paths from a to d.

Prolog Database Representation. Datalog Convention.

Library: an example database structure:

1 book(signature_1,title_1,author_1,publisher_1,year_1).
2 book(signature_2,title_2,author_2,publisher_2,year_2).
3 book(signature_3,title_3,author_1,publisher_3,year_3).
4 book(signature_4,title_4,author_4,publisher_4,year_4).
5 book(signature_5,title_5,author_1,publisher_5,year_5).
6

7 journal(journal_id_1,title_1,volume_1,number_1,year_1).
8 journal(journal_id_2,title_2,volume_2,number_2,year_2).
9 journal(journal_id_3,title_3,volume_3,number_3,year_3).
10 journal(journal_id_4,title_4,volume_4,number_4,year_4).
11 journal(signature_5,title_5,volume_5,number_5,year_5).
12

13 user(id_1,surname_1,forename_1,born_1,address_1).
14 user(id_2,surname_2,forename_2,born_2,address_2).
15 user(id_3,surname_3,forename_3,born_3,address_3).
16 user(id_4,surname_4,forename_4,born_4,address_4).
17 user(id_5,surname_5,forename_5,born_5,address_5).
18

19 register(id_1,signature_1,borrow_date_1,return_date_1).
20 register(id_1,signature_2,borrow_date_2,return_date_2).
21 register(id_2,signature_1,borrow_date_3,return_date_3).

Prolog Database Representation. Datalog Convention.

Library: example database operations

1 sum(Id,Title,Year):-
2 book(Id,Title,_,_,Year);
3 journal(Id,Title,_,_,Year), not(book(Id,Title,_,_,Year)).
4

5 intersect(Id,Title,Year):-
6 book(Id,Title,_,_,Year),
7 journal(Id,Title,_,_,Year).
8

9 except(Id,Title,Year):-
10 book(Id,Title,_,_,Year),
11 not(journal(Id,Title,_,_,Year)).
12

13 projection(Title,Author,Year):-
14 book(_,Title,Author,_,Year).
15

16 selection(Signature,Title,Author,Publisher,Year):-
17 book(Signature,Title,Author,Publisher,Year),
18 Author = author_1.
19

20 cartesian_product(Signature,Uid):-
21 book(Signature,_,_,_,_),
22 user(Uid,_,_,_,_).

Prolog Database Representation. Datalog Convention.

Library: example database operations

1 inner_join(Uid,Surname,Name,Signature):-
2 user(Uid,Surname,Name,_,_),
3 register(Uid,Signature,_,_).
4

5 left_outer_join(Uid,Surname,Name,Signature):-
6 user(Uid,Surname,Name,_,_),
7 register(Uid,Signature,_,_).
8 left_outer_join(Uid,Surname,Name,Signature):-
9 user(Uid,Surname,Name,_,_),
10 not(register(Uid,Signature,_,_)).
11

12 who_what_book(Uid,Surname,Name,Signature,
13 Title,BorrowDate,ReturnDate):-
14 register(Uid,Signature,BorrowDate,ReturnDate),
15 user(Uid,Surname,Name,_,_),
16 book(Signature,Title,_,_,_).
17

18 subquery_has_book(Uid,Surname,Name):-
19 user(Uid,Surname,Name,_,_),
20 register(Uid,_,_,_).

Deductive Databases in PROLOG

1 rodzina(osoba(jan, kowalski,data(5,kwiecien,1946),pracuje(tpsa,3000)),
2 osoba(anna,kowalski,data(8,luty,1949), pracuje(szkola,1500)),
3 [osoba(maria,kowalski,data(20,maj,1973), pracuje(argo_turist,4000)),
4 osoba(pawel,kowalski,data(15,listopad,1979),zasilek)]).
5

6 rodzina(osoba(krzysztof, malinowski, data(24,lipiec,1950), bezrobocie),
7 osoba(klara, malinowski, data(9,styczen,1951), pracuje(kghm,8000)),
8 [osoba(monika, malinowski, data(19,wrzesien,1980), bezrobocie)]).
9

10 maz(X) :- rodzina(X,_,_).
11 zona(X) :- rodzina(_,X,_).
12 dziecko(X) :- rodzina(_,_,Dzieci), member(X,Dzieci).
13

14 istnieje(Osoba) :- maz(Osoba); zona(Osoba); dziecko(Osoba).
15

16 data_urodzenia(osoba(_,_,Data,_),Data).
17

18 pensja(osoba(_,_,_,pracuje(_,P)),P).
19 pensja(osoba(_,_,_,zasilek),500).
20 pensja(osoba(_,_,_,bezrobocie),0).
21

22 zarobki([],0).
23 zarobki([Osoba|Lista],Suma) :-
24 pensja(Osoba,S),
25 zarobki(Lista,Reszta),
26 Suma is S + Reszta.

Inference Control Predicates in PROLOG

Built-in Predicates
1 fail – always fails. Since it is impossible to pass through, it is used to force

backtracking.
2 true – always succeeds.
3 repeat – always succeeds; provides an infinite number of choice points.
4 ! – cut; prohibits backtracking to the goals located left of its placement.
5 +Goal1, +Goal2 – conjunction; prove Goal1, then Goals2.
6 +Goal1; +Goal2 – disjunction; either Goal1 or Goal2 should be

proved.It is translated into:

Goal12 :- Goal1.
Goal12 :- Goal2.

Disjunction in clauses:

h :- p;q.

can be translated into:

h :- p.
h :- q.

Definition of repeat
repeat.
repeat:- repeat.

The Role and Use of fail

Role of fail

The fail predicate does not unify with any other predicate.
1 fail — stops further inference, and
2 enforces backtracking.

Applications of fail

1 to ensure exploration of all the possible executions of a clause; fail is placed
at the end of this clause.

2 definition of repeated actions (a loop),
3 ensuring program execution (robust programming).

An action-fail loop

1 loop :-
2 action,
3 fail.
4 loop.

A repeat-test loop

1 loop :-
2 repeat,
3 action,
4 test.
5 test :- termination_ok,!,go_out.
6 test :- fail.

Loops-repeat-fail

Example loop solutions

1 loop_infinite:-
2 repeat,
3 actions,
4 fail.
5

6 loop_infinite_read_write:-
7 repeat,
8 read(X), process(X,Y), write(Y),nl,
9 fail.
10

11 loop_find_fail:-
12 d(X),
13 process(X,Y), write(Y),nl,
14 fail.
15 loop_find_fail.
16

17 loop_repeat_test:-
18 repeat,
19 d(X),
20 process(X,Y), write(Y),nl,
21 test_for_final(Y), write(’***end: ’),write(Y),nl.
22 d(0). d(1). d(2). d(3). d(4). d(5). d(6). d(7). d(8). d(9).

The cut predicate

Description of cut

z The cut predicate is a standard predicate which helps avoiding exploration of
further inference possibilities, provided that the one currently explored satisfies
certain predicates.

z The cut predicate allows for pruning branches in the search tree generated by
the depth-first search algorithm.

z The cut predicate is symbolized with the exclamation mark ‘!’.

Definition and operation of cut

Consider a clause:
h :- p1, p2, . . . , pi, !, pi+1, . . . , pm.

1 ! divides a clause in the left and right part,
2 the left part atoms are removed form stack,
3 no backtracking for h, p1, p2, . . . , pi,
4 backtracking is still possible for pi+1, . . . , up to pm.

Use of Cut: Example

Examples with cut

1 cyfra(0).
2 cyfra(1).
3 cyfra(2).
4 cyfra(3).
5 cyfra(4).
6 cyfra(5).
7 cyfra(6).
8 cyfra(7).
9 cyfra(8).
10 cyfra(9).
11

12 liczba(X):- %%% 0-99
13 cyfra(S),!,
14 cyfra(D),
15 cyfra(J),
16 X is 100*S+10*D+J.
17

18 liczba(X):- %%% 0-9
19 cyfra(S),
20 cyfra(D),!
21 cyfra(J),
22 X is 100*S+10*D+J.

Use of Cut

Example: definition of a function

1 %%% Function definition: mutually exclusive conditions
2

3 f(X,2):- X>=0,X<3.
4 f(X,4):- X>=3,X<6.
5 f(X,6):- X>=6,X<9.
6 f(X,8):- X>=9.
7

8 %%% Function definition: preventing backtrackin with cut
9

10 fc(X,2):- X>=0,X<3,!.
11 fc(X,4):- X>=3,X<6,!.
12 fc(X,6):- X>=6,X<9,!.
13 fc(X,8):- X>=9.
14

15 %%% Function definition: preventing backtracking + optimization
16

17 fco(X,2):- X>=0,X<3,!.
18 fco(X,4):- X<6,!.
19 fco(X,6):- X<9,!.
20 fco(X,8):- X>=9.

Negation in PROLOG

Closed World Assumption

z positive knowledge about the world of interest is stated explicitly; it takes the
form of facts and rules (clauses in PROLOG),

z it is assumed that all the positive knowledge is available or can be deduced; in
other words, that the world is closed,

z if so, if a certain fact does not follow from the current knowledge base, it is
assumed to be false,

z it does not mean it is really false; the definition is operational — it provides a
way to decide whether something is false in the case the negation of it is not
stated in an explicit way.

Basic concept of negation in PROLOG

1 Negation in PROLOG is based on the Closed World Assumption,
2 it is implemented as negation as failure to prove a goal,
3 predicate not is in fact a meta-predicate — its argument is a predicate p, and

not(p) succeeds if an attempt to prove p fails,
4 not(p) — attempts to prove p; if the interpreter fails, not(p) succeeds.

Syntax

Example use of not

1 %%% Large base of positive facts
2 pracownik(adam).
3 pracownik(bogdan).
4 pracownik(czesiek).
5 pracownik(damian).
6 pracownik(eustachy).
7 pracownik(walery).
8 pracownik(xawery).
9

10 %%% Few exception examples
11 bumelka(bogdan).
12 bumelka(walery).
13 bumelka(damian).
14

15 premia(X):-
16 pracownik(X),
17 not(bumelka(X)).
18

19 %%% Variable instantiation first!!!
20 premia_wrong_example(X):-
21 not(bumelka(X)),
22 pracownik(X).

Syntax

Implementation of not

1 not(P):- call(P),!,fail.
2 not(_).

Use of not

not(goal) should be always invoked with all variables of the goal instantiated!

Example use of not

1 %%% Variable instantiation first!!!
2 premia_wrong_example(X):-
3 not(bumelka(X)),
4 pracownik(X).
5 %%% Explanation of the work of not
6 pracowity(X):-
7 bumelka(X),!,
8 fail.
9 pracowity(_).

10

11 w:- pracownik(X),write(X),nl, pracowity(X), write(X),nl,fail.

Global Memory in PROLOG

Asserting and Retracting knowledge

1 Programs in PROLOG can easily access global memory.
2 There are two basic operations on it — one can assert new facts and the other

can retract them.
3 The facts can be read from any place of the program.
4 The standard predicates assert/1 and retract/1 are in fact meta-predicates; their

arguments are facts.
5 practical use:

assert(p).

retract(p).

6 The retract and assert operations produce results which are not removed during
backtracking.

7 Predicates assert and retract can be used in any place of the program
8 At any stage of program execution all the clauses have access to the knowledge

contained in the global memory; a useful mechanism for communication among
clauses without parameter passing is available in this way.

Global Memory

Examples

1 assert(p(a)).
2 assert(p(a,f(b),g(a,c))).
3 assert(p(X,X)).
4 assert(list,[a,b,c,d,e],5).
5 assert(register(id_1,signature_1,borrow_date_1,return_date_1)).

Cleaning the Database

The predicate which performs clearing of the memory is called retractall/1; its
definition is as follows:

retractall(P):-
retract(P),
fail.

retractall(_).

retractall/1 is built-in predicate in SWI-PROLOG.

Buil-in predicates

Dynamic facts declaration

Dynamic facts should be declared as such

1 :- dynamic p/n.

Database predictaes

1 dynamic p/n – declared a dynamic term p with arity n.
2 consult(+File) – reads the file ’File’ into the program (A Prolog source

code; usually a dynamic database).
3 assert(+Term) – asserts Term to the database (at the last position).
4 asserta(+Term) – asserts Term to the database (at the first position).
5 assertz(+Term) – asserts Term to the database (at the last position).
6 retract(+Term) – retracts the first term unifiable with Term.
7 retractall(+Term) – retracts all the occurrences of terms unifiable with
Term.

8 abolish(Functor/Arity) – Removes all clauses of a predicate with
functor Functor and arity Arity from the database.

9 abolish(+Name, +Arity) – the same as abolish(Name/Arity).

Syntax

1 :- op( 50, xfy, :).
2 route(P1, P2, Day, [P1 / P2 / Fnum / Deptime]):- % Direct flight
3 flight( P1, P2, Day, Fnum, Deptime, _).
4 route(P1, P2, Day, [(P1 / P3 / Fnum1 / Dep1)|RestRoute] ):- %Indirect
5 route( P3, P2, Day, RestRoute),
6 flight( P1, P3, Day, Fnum1, Dep1, Arr1),
7 deptime( RestRoute, Dep2), % Departure time of Route
8 transfer( Arr1, Dep2). % Enough time for transfer
9 flight( Place1, Place2, Day, Fnum, Deptime, Arrtime) :-
10 timetable( Place1, Place2, Flightlist),
11 member( Deptime / Arrtime / Fnum / Daylist , Flightlist),
12 flyday( Day, Daylist).
13 flyday( Day, Daylist) :-
14 member( Day, Daylist).
15 flyday( Day, alldays) :-
16 member( Day, [mo,tu,we,th,fr,sa,su] ).
17 deptime( [ _ / _ / _ / Dep | _], Dep).
18 transfer( Hours1:Mins1, Hours2:Mins2) :-
19 60 * (Hours2 - Hours1) + Mins2 - Mins1 >= 40.
20 % A FLIGHT DATABASE
21 timetable( edinburgh, london,
22 [ 9:40 / 10:50 / ba4733 / alldays,
23 13:40 / 14:50 / ba4773 / alldays,
24 19:40 / 20:50 / ba4833 / [mo,tu,we,th,fr,su] ] ).
25 timetable( london, edinburgh,
26 [ 9:40 / 10:50 / ba4732 / alldays,
27 11:40 / 12:50 / ba4752 / alldays,
28 18:40 / 19:50 / ba4822 / [mo,tu,we,th,fr] ] ).
29 timetable( london, ljubljana,
30 [ 13:20 / 16:20 / jp212 / [mo,tu,we,fr,su],
31 16:30 / 19:30 / ba473 / [mo,we,th,sa] ] ).
32 timetable( london, zurich,
33 [ 9:10 / 11:45 / ba614 / alldays,
34 14:45 / 17:20 / sr805 / alldays ] ).
35 timetable( london, milan,
36 [ 8:30 / 11:20 / ba510 / alldays,
37 11:00 / 13:50 / az459 / alldays ] ).
38 timetable( ljubljana, zurich,
39 [ 11:30 / 12:40 / jp322 / [tu,th] ] ).
40 timetable( ljubljana, london,
41 [ 11:10 / 12:20 / jp211 / [mo,tu,we,fr,su],
42 20:30 / 21:30 / ba472 / [mo,we,th,sa] ] ).
43 timetable( milan, london,
44 [ 9:10 / 10:00 / az458 / alldays,
45 12:20 / 13:10 / ba511 / alldays ] ).
46 timetable( milan, zurich,
47 [ 9:25 / 10:15 / sr621 / alldays,
48 12:45 / 13:35 / sr623 / alldays ] ).
49 timetable( zurich, ljubljana,
50 [ 13:30 / 14:40 / jp323 / [tu,th] ] ).
51 timetable( zurich, london,
52 [ 9:00 / 9:40 / ba613 / [mo,tu,we,th,fr,sa],
53 16:10 / 16:55 / sr806 / [mo,tu,we,th,fr,su] ] ).
54 timetable( zurich, milan,
55 [ 7:55 / 8:45 / sr620 / alldays ] ).

The Zebra Puzzle — Einstein Problem

1 left(L,P,[L,P,_,_,_]).
2 left(L,P,[_,L,P,_,_]).
3 left(L,P,[_,_,L,P,_]).
4 left(L,P,[_,_,_,L,P]).
5

6 near(X,Y,L):- left(X,Y,L).
7 near(X,Y,L):- left(Y,X,L).
8

9 einstein(S):-
10 S = [[norweg,_,_,_,_],_,[_,_,_,mleko,_],_,_],
11 member([anglik,czerwony,_,_,_],S),
12 member([szwed,_,psy,_,_],S),
13 member([dunczyk,_,_,herbata,_],S),
14 left([_,zielony,_,_,_],[_,bialy,_,_,_],S),
15 member([_,zielony,_,kawa,_],S),
16 member([_,_,ptaki,_,pallmall],S),
17 member([_,zolty,_,_,dunhill],S),
18 near([_,_,_,_,blends],[_,_,koty,_,_],S),
19 near([_,_,konie,_,_],[_,_,_,_,dunhill],S),
20 member([_,_,_,piwo,blumaster],S),
21 member([niemiec,_,_,_,prince],S),
22 near([norweg,_,_,_,_],[_,niebieski,_,_,_],S),
23 near([_,_,_,_,blends],[_,_,_,voda,_],S),
24 member([_,_,rybki,_,_],S).

